

Geburtstagstermine mit MS OUTLOOK überwachen

Dr. Volker Thormählen, 09. Augst 2019

Inhaltsverzeichnis

Inhaltsverzeichnis.....	I
1 Aufgabe	1
2 Verweise setzen	1
3 In OUTLOOK eingebaute Funktionalität	1
4 Verschieben von Serienterminen in den Geburtstagskalender	2
5 Geburtstagskalender neu erstellen	2
6 Kategorie neu erstellen	3
7 Geburtstagskalender mit VBA-Prozeduren automatisch füllen.....	4
7.1 Datenquelle ist ein EXCEL-Tabellenblatt.....	4
7.2 Datenquelle sind OUTLOOK-Kontaktdaten.....	4
8 Inhalt des Geburtstagskalenders	6
8.1 <i>Inhalt anzeigen</i>	6
8.2 <i>Inhalt entfernen</i>	8
9 Geburtstagsgrüße automatisch erzeugen	8
9.1 <i>E-Mail-Entwurf speichern</i>	8
9.2 <i>VBA-Prozedur 'BirthdayGreetings' starten</i>	9
9.3 Geburtstagsgrüße versenden	10
10 VBA-Prozeduren	II
10.1 Serientermine in den Geburtstagskalender verschieben.....	II
10.2 Geburtstagstermine von EXCEL-Tabellenblatt importieren.....	III
10.3 Geburtstagstermine aus OUTLOOK-Kontaktdaten importieren	V
10.4 Alle Geburtstagstermine aus dem Geburtstagskalender entfernen	VII
10.5 Geburtstagsgrüße erzeugen	VIII
11 Verzeichnisse.....	X
11.1 Abbildungen	X
11.2 Listings.....	X
11.3 Literatur.....	X

1 Aufgabe

In diesem Beitrag werden zwei Möglichkeiten beschrieben, wie Geburtsdaten in einen OUTLOOK¹-Geburtskalender importiert und dort terminlich überwacht werden können:

- Import aus einem EXCEL-Tabellenblatt
- Import aus den OUTLOOK-Kontakten

In beiden Fällen sollte in OUTLOOK neben dem Standard- ein Geburtskalender vorhanden sein. Sonst muss letzterer neu erstellt werden.

Alle Schritte können schnell und sicher mit EXCEL-Makros durchgeführt werden:

- Der Import der Geburtsdaten aus dem EXCEL-Tabellenblatt erfolgt mit der VBA²-Prozedur *'ImportBirthdatesToCalendar'* (s. Listing 2)
- Der Import der Geburtsdaten aus den OUTLOOK-Kontakten wird mit der VBA-Prozedur *'CreateBirthdayEventsFromContacts'* durchgeführt (s. Listing 3).

Die Erstellung eines Geburtskalenders wird in OUTLOOK am einfachsten manuell vorgenommen. Für das Verschieben von Serienterminen zwischen definierten Kalendern in OUTLOOK werden verschiedene Vorgehensweisen beschrieben (s. Seite 2).

2 Verweise setzen

Alle präsentierten VBA-Prozeduren werden von EXCEL aus gestartet. Deshalb müssen dort die Verweise wie folgt gesetzt werden (s. Abbildung 1):




Abbildung 1: Verweise setzen zur Office Automation

3 In OUTLOOK eingebaute Funktionalität

Wenn ein neuer Kontakt in OUTLOOK erstellt und dafür ein Geburtsdatum eingegeben wird, erzeugt OUTLOOK automatisch einen Serientermin für den Geburtstag im Standard-/Hauptkalender und erinnert – wenn gewünscht – an diesen Termin. Das gilt auch für den Fall, dass bei einem schon erfassten Kontakt das zugehörige Geburtsdatum nachträglich eingegeben oder geändert wird. Die Erinnerung an einen Termin durch ein Signal von OUTLOOK ist optional.

¹ Outlook eine proprietäre Software der Firma Microsoft zum Empfangen und Versenden von E-Mails sowie zum Verwalten von Terminen, Kontakten, Aufgaben und Notizen. Sie läuft unter dem Betriebssystem Windows.

² VBA ist ein Akronym für **V**isual **B**asic for **A**pplications

Um Geburtstagstermine *nicht* im Standard-/Hauptkalender zu speichern, sondern in einem gesonderten Kalender, bietet sich die Erstellung eines Geburtstagskalenders an. Auf diese Weise lassen sich private und berufliche/geschäftliche Termine getrennt anzeigen, überwachen, entfernen, stillstellen, etc.

4 Verschieben von Serienterminen in den Geburtstagskalender

Das Verschieben von ausgewählten Elementen des Standard-/Hauptkalenders in einen vorhandenen Geburtstagskalender kann in OUTLOOK mit der eingebauten Funktion 'Verschieben' erfolgen (s. Abbildung 2):




Abbildung 2: Schaltfläche zum Verschieben von Kalenderelementen

Im Ergebnis werden alle ausgewählten Serientermine des Standard-/Hauptkalenders schnell und einfach in den Geburtskalender verschoben oder umgekehrt.

Alternativ kann das Verschieben von Serienterminen zwischen definierten Kalendern mit der Prozedur 'MoveAppointments' durchgeführt werden (s. Listing 1 und Abbildung 3):




Abbildung 3: Erfolgsmeldung nach dem Verschieben von Serienterminen

Das *automatische* Verschieben zwischen Kalendern kann ebenfalls mit einer VBA-Prozedur vorgenommen werden, die in 'ThisOutlookSession' installiert werden muss. (siehe hier [1]).

5 Geburtstagskalender neu erstellen

Wenn nicht bereits ein Geburtstagskalender vorhanden ist, wird dieser mit folgendem Menüpfad³ erstellt: *Kalender > Ordner > Neu > Neuer Kalender*. Das daraufhin erscheinende Formular mit dem Namen 'Neuen Ordner erstellen' (s. Abbildung 4) wird mit 'Geburtstagskalender' belegt und danach mit OK bestätigt⁴:

³ MS Outlook 2010

⁴ In Outlook selbst ist gewöhnlich kein Geburtstagskalender vorhanden, da Geburtstage automatisch in den Standard-/Hauptkalender eingefügt werden, wenn das Geburtsdatum für einen neuen Kontakt eingegeben wird.




Abbildung 4: Geburtstagskalender in OUTLOOK neu erstellen

6 Kategorie neu erstellen

Eine OUTLOOK-Kategorie ist ein Attribut, das jedem beliebigen OUTLOOK-Element zugeordnet werden kann. Anhand dieses Attributs können die verschiedenen OUTLOOK-Elemente (Kalendereinträge, Termine, Aufgaben, Kontaktadressen, Notizen, E-Mails etc.) beliebig sortiert, gruppiert, angezeigt oder gesucht werden.

Die Erstellung einer neuen, benutzerdefinierten Kategorie kann mit der Navigation *Start > Kategorien > Kategorisieren* eingeleitet werden (s. Abbildung 5)




Abbildung 5: Schaltfläche Kategorisieren im OUTLOOK-Menüband

In dem sich öffnenden Drop-down Menü wird der Befehl 'Alle Kategorien' und danach die Option 'Neu' ausgewählt. Die angezeigten Formularfelder werden wie gewünscht belegt (s. Abbildung 6) und mit 'OK' bestätigt.




Abbildung 6: Neue Kategorie 'Geburtstag' hinzufügen

Die Kategorie 'Geburtstag' ist damit in OUTLOOK verfügbar.

7 Geburtstagskalender mit VBA-Prozeduren automatisch füllen

Datenquelle für das Einfügen von Geburtstagsdaten ist entweder eine EXCEL-Tabelle oder OUTLOOK-Kontakte. Die dafür notwendigen VBA-Prozeduren wurden bereits weiter oben (s. Seite 1) erwähnt.

7.1 Datenquelle ist ein EXCEL-Tabellenblatt

Bei Verwendung einer EXCEL-Tabelle könnten die Geburtstagsdaten beispielsweise so aussehen (s. Abbildung 7):

	A	B
1	Name	Geburtsdatum
2	Annette	18.3.1947
3	Cord	15.6.1947
4	Thies	27.8.1944
5	Thorsten	6.11.1973
6	Volker	14.9.1940

Abbildung 7: Geburtstagstermine in EXCEL-Tabellenblatt

Die Erfolgsmeldung der VBA-Prozedur 'ImportBirthdatesToCalendar' (s. Listing 2) lautet dementsprechend (s. Abbildung 8):




Abbildung 8: Geburtstagstermine in OUTLOOK-Geburtstagskalender eintragen

7.2 Datenquelle sind OUTLOOK-Kontaktdaten

Geburtstagsdaten aus den OUTLOOK-Kontaktdaten können sinnvoller Weise nur importiert werden, wenn das entsprechende Feld ein gültiges Geburtsdatum enthält. Die *allgemeinen* Kontaktdaten enthalten aber *kein* solches Formularfeld (s. Abbildung 9). Erst durch einen Klick auf die Schaltfläche *Details* (s. Abbildung 10) wird das gesuchte Formularfeld sichtbar (s. Abbildung 11).

Datei	Kontakt	Einfügen	Text formatieren	Überprüfen	Entwicklertools
Speichern & schließen	Löschen	Speichern und neuer Kontakt ▾	Weiterleiten ▾	OneNote	Aktivitäten
Aktionen				Anzeigen	
Name...: Volker Thormählen Firma: Position: Speichern unter: Thormählen, Volker ▾					
Internet					
E-Mail... ▾: volker@dr-thormaehlen.de Anzeigen als: Volker Thormählen (volker@dr-thormaehlen.de) Webseite: www.dr-thormaehlen.de Chatadresse:					
Telefonnummern					
Geschäftlich... ▾: Privat... ▾: 0210252141 Fax geschäftl.... ▾: Mobiltelefon... ▾: 01737935232					
Adressen					
Privat... ▾: Steinhauser Str. 54 <input type="checkbox"/> Dies ist die Postanschrift		40882 Ratingen NRW		Zuordnen	

Abbildung 9: OUTLOOK-Kontaktdaten von Volker Thormählen

Allgemein	Details	Aktivitäten
		Zertifikate
		Alle Felder
Anzeigen		

Abbildung 10: Details eines OUTLOOK-Kontakts erfassen

Datei	Kontakt	Einfügen	Text formatieren	Überprüfen	Entwicklertools
Speichern & schließen	Löschen	Speichern und neuer Kontakt	Weiterleiten	OneNote	Details
Aktionen			Anzeigen		Komm
Abteilung:			Vorgesetzte(r):		
Büro:			Assistent:		
Beruf:					
Spitzname:			Partner/in:		
Anrede:			Geburtstag:		
Namenszusatz:			Jahrestag:		
Internet Frei/Gebucht					
Adresse:					

Abbildung 11: Feld zur Aufnahme des Geburtstags des zuvor gezeigten OUTLOOK-Kontakts

Wenn in den OUTLOOK-Kontaktdaten gültige Geburtstagsdaten gefunden werden, dann gibt die VBA-Prozedur 'CreateBirthdayEventsFromContacts' (s. Listing 3) folgende Meldung aus (s. Abbildung 12):




Abbildung 12: Geburtstage in OUTLOOK-Kontakte finden und in den Geburtstagskalender einstellen

8 Inhalt des Geburtstagskalenders

8.1 Inhalt anzeigen

Im linken Bereich auf *Geburtskalender* klicken (direkt unter dem Symbol des *Standardkalenders*) (s. Abbildung 13):



Abbildung 13: Geburtstagskalender öffnen

Über den Menüpfad *Ansicht > Aktuelle Ansicht > Ansicht ändern > Liste* kann die Liste der Einträge im Geburtstagskalender angezeigt werden (s. Abbildung 14):

Betreff	Termin-/Besprechungsort	Beginn	Ende	Serienmuster	Kategorien
Kategorien: Geburtstag (5 Elemente)					
Geburtstag von Annette	[Lebensalter: 71]	Di 18.03.1947 00:00	Mi 19.03.1947 00:00	jedes Jahr am 18. März	Geburtstag
Geburtstag von Cord	[Lebensalter: 71]	So 15.06.1947 00:00	Mo 16.06.1947 00:00	jedes Jahr am 15. Juni	Geburtstag
Geburtstag von Volker	[Lebensalter: 78]	Sa 14.09.1940 00:00	So 15.09.1940 00:00	jedes Jahr am 14. September	Geburtstag
Geburtstag von Thies	[Lebensalter: 74]	So 27.08.1944 00:00	Mo 28.08.1944 00:00	jedes Jahr am 27. August	Geburtstag
Geburtstag von Thorsten	[Lebensalter: 45]	Di 06.11.1973 00:00	Mi 07.11.1973 00:00	jedes Jahr am 6. November	Geburtstag

Abbildung 14: Liste der Einträge im OUTLOOK-Geburtstagskalender

Das in Abbildung 14, Spalte *Termin-/Besprechungsort*, enthaltene Lebensalter wird mit der Funktion 'LengthOfLife' (s. Listing 5) bestimmt.

Bei einem Klick auf das Symbol links neben dem Betreff 'Geburtstag von Volker' (s. Abbildung 14) werden die Details der dazugehörigen Terminserie angezeigt (s. Abbildung 15):

Terminserie

Termin

Beginn: 00:00

Ende: 00:00

Dauer: 1 Tag

Serienmuster

☐ Täglich

☐ Wöchentlich

☐ Monatlich

☒ Jährlich

Jedes/Alle 1 Jahr(e)

☒ Am: September . 14

☐ Am: zweiten Samstag im September

Seriendauer

Beginn: Sa 14.09.1940

☒ Kein Enddatum

☐ Endet nach: 10 Terminen

☐ Endet am: Mi 14.09.1949

OK

Abbrechen

Serie entfernen

Abbildung 15: Terminserie für den OUTLOOK-Kontakt 'Volker Thormählen'

8.2 Inhalt entfernen

Bei einem Klick auf die Kategorie *'Geburtstag (5 Element)'* (siehe Abbildung 14, oben) können alle fünf Einträge des Geburtstagskalenders entfernt werden.

Alternativ können alle Elemente des Geburtstagskalenders mit der VBA-Prozedur *'RemoveBirthdayEvents'* (s. Listing 6) entfernt werden. Die zugehörige Erfolgsmeldung lautet (s. Abbildung 16):




Abbildung 16: Erfolgsmeldung bei Entfernung aller Termine aus dem Geburtstagskalender

9 Geburtstagsgrüße automatisch erzeugen

9.1 E-Mail-Entwurf speichern

Angestoßen durch eine Erinnerung von OUTLOOK an einen Geburtstag kann ein E-Mail-Entwurf mit entsprechenden Grüßen automatisch erzeugt werden. Voraussetzung dafür ist die Existenz eines entsprechenden E-Mail-Entwurfs mit Betreff *„Template“* (siehe Abbildung 1).



Abbildung 17: E-Mail-Entwurf unter dem Namen *'Template'* speichern

Ein solcher E-Mail-Entwurf muss nur 1-mal erstellt und kann immer wieder für den vorgesehenen Zweck verwendet werden. Um Erfassungsaufwand zu sparen, sollte er deshalb nicht aus den sog. *'Entwürfen'* entfernt werden.

Wird der Name eines solchen E-Mail-Entwurf von der VBA-Prozedur (siehe Gliederungspunkt 9.2) *nicht* gefunden, wird trotzdem ein neuer, versandfertiger E-Mail-Entwurf mit *"Herzlichen Glückwunsch zum Geburtstag!"* im Textkörper erzeugt.

Bei fehlerfreiem Ablauf gibt die VBA-Prozedur diese Erfolgsmeldung aus (s. Abbildung 18):




Abbildung 18: Erfolgsmeldung der Prozedur 'BirthdayGreetings'

9.2 VBA-Prozedur 'BirthdayGreetings' starten

Um den Entwurf des jeweiligen Geburtstagsgrußes automatisch zu erzeugen, muss die VBA-Prozedur 'BirthdayGreetings' gestartet werden (s. Listing 7). Diese erwartet als Eingabe das gültige Geburtsdatums des jeweiligen OUTLOOK-Kontakts, sonst bricht sie mit einer entsprechenden Fehlermeldung ab.




Abbildung 19: Geburtsdatum eines OUTLOOK-Kontakts erfassen

Die genannte VBA-Prozedur erstellt einen neuen E-Mail-Entwurf auf der Basis des weiter oben beschriebenen Entwurfs (siehe Gliederungspunkt 9.1). Wird dieser Entwurf nicht gefunden, wird trotzdem ein neuer E-Mail-Entwurf generiert, allerdings mit dem lapidaren Text "Herzlichen Glückwunsch zum Geburtstag!" im Textkörper (engl. 'body').

Letztlich sieht der automatisch erzeugte neue E-Mail-Entwurf wie folgt aus (s. Abbildung 20



Abbildung 20: Neuer Entwurf einer E-Mail mit Geburtstagsgrüßen

9.3 Geburtstagsgrüße versenden

Der gespeicherte neue E-Mail-Entwurf kann nun versendet werden, eventuell nach einer Überarbeitung. Nach dem Senden wird er von OUTLOOK automatisch gelöscht.

10 VBA-Prozeduren

10.1 Serientermine in den Geburtstagskalender verschieben

```
Sub MoveAppointments()  
    ' Aufgabe: Alle Serientermine verschieben vom Hauptkalender in den Geburtstagskalender.  
    Const conMsg As String = " Serientermin(e) in den Geburtstagskalender verschoben!"  
    Dim OL As Outlook.Application  
    Dim olAppt As Outlook.AppointmentItem  
    Dim objDefaultCal As Object  
    Dim objTargetCal As Object  
    Dim bolCreated As Boolean  
    Dim intCnt As Integer  
    Dim intLoop As Integer  
    ' Outlook starten  
    On Error Resume Next  
    Set OL = GetObject(, "Outlook.Application")  
    bolCreated = True  
    If OL Is Nothing Then  
        Set OL = CreateObject("Outlook.Application")  
        bolCreated = False  
    End If  
    ' Fehlerbehandlungsroutine setzen  
    On Error GoTo Error_Point  
    Set objDefaultCal = OL.Session.GetDefaultFolder(olFolderCalendar) ' Standard-/Hauptkalender  
    Set objTargetCal = objDefaultCal.Folders("Geburtstagskalender") ' Geburtstagskalender  
    ' Zahl der Serientermine im Hauptkalender  
    intCnt = objDefaultCal.Items.Count  
    If intCnt > 0 Then  
        For intLoop = 1 To intCnt Step 1  
            Set olAppt = objDefaultCal.Items(intLoop)  
            olAppt.IncludeRecurrences = True  
            If (TypeOf olAppt Is AppointmentItem) And _  
                (olAppt.IsRecurring = True) And _  
                (Left(olAppt.Subject, 10) = "Geburtstag") And _  
                olAppt.Categories = "Geburtstag" Then  
                ' Serientermin in den Geburtstagskalender verschieben  
                olAppt.Move objTargetCal  
            End If  
        Next intLoop  
        MsgBox Str(intCnt) & conMsg, vbInformation, "Serientermine verschieben"  
    Else  
        MsgBox "Keine " & conMsg, vbInformation, "Serientermine verschieben"  
    End If  
Exit_Point:  
    On Error Resume Next  
    If bolCreated Then OL.Quit  
    ' Objektvariablen freigeben  
    Set OL = Nothing  
    Set olAppt = Nothing  
    Set objDefaultCal = Nothing  
    Set objTargetCal = Nothing  
    ' Prozedur sofort verlassen  
Exit Sub  
Error_Point:  
    Resume Exit_Point  
End Sub
```

Listing 1: Serientermine in den Geburtstagskalender verschieben

10.2 Geburtstagstermine von EXCEL-Tabellenblatt importieren

```

Sub ImportBirthdatesToCalendar()
' Aufgabe: Geburtstagstermine von EXCEL nach OUTLOOK importieren
Dim objWS As Excel.Worksheet
Dim objOL As Object
Dim olAppt As Outlook.AppointmentItem
Dim olFolder As Outlook.Folder
Dim olPattern As Outlook.RecurrencePattern
Dim bolCreated As Boolean
Dim dtmBirthdate As Date
Dim lngHits As Long
Dim lngLastRow As Long
Dim lngRow As Long
Dim strAge As String
Dim strFirstName As String
' Fehlerbehandlungsroutine setzen
On Error GoTo Error_Point

' EXCEL-Arbeitsblatt bestimmen
Set objWS = ThisWorkbook.Sheets(2)
' letzte Zeile im betreffenden EXCEL-Arbeitsblatt bestimmen
lngLastRow = objWS.Range("A" & objWS.Rows.Count).End(xlUp).Row

' OUTLOOK starten
On Error Resume Next
Set objOL = GetObject(, "Outlook.Application")
If Err.Number <> 0 Then
    Err.Clear
    Set objOL = CreateObject("Outlook.Application")
    bolCreated = True
Else
    bolCreated = False
End If

' OUTLOOK-Geburtstagskalender verwenden
Set olFolder = objOL.Session.GetDefaultFolder(olFolderCalendar).Folders("Geburtstagskalender")

' Schleife über alle Zeilen im EXCEL-Arbeitsblatt (ohne 1. Zeile)
For lngRow = 2 To lngLastRow
    With objWS
        strFirstName = objWS.Range("A" & lngRow)
        dtmBirthdate = FormatDateTime(objWS.Range("B" & lngRow), vbShortDate)
    End With
    ' Mit 'CheckCalEntry' prüfen, ob d. zugehörige Kalendereintrag noch nicht vorhanden ist.
    If Not CheckCalEntry(dtmBirthdate, strFirstName, objOL) Then
        Set olAppt = olFolder.Items.Add("IPM.Appointment")
        ' Geburtstags-Termin erstellen
        With olAppt
            ' Ganztägiges Ereignis
            .AllDayEvent = True
            ' Betreff
            .Subject = "Geburtsdatum von " & strFirstName
            ' Vorlaufzeit in Minuten zwecks Erinnerung an d. Termin/Ereignis
            .ReminderMinutesBeforeStart = 1440 ' 24 Stunden
            ' Schalter zur Erinnerung setzen
            .ReminderSet = True
            ' Kategorie zuordnen
            .Categories = "Geburtstag"
            ' Vertraulichkeit
            .Sensitivity = olPrivate
            ' Lebensalter in Jahren berechnen mit der Funktion 'LengthOfLife'
            strAge = LengthOfLife(dtmBirthdate, Now())
            .Location = "[Lebensalter: " & strAge & "]"
            ' Muster für Terminserie festlegen
            Set olPattern = .GetRecurrencePattern
        End With
        With olPattern
            .RecurrenceType = olRecurYearly ' Häufigkeit: 1-mal jährlich
            .PatternStartDate = dtmBirthdate ' Startdatum
            .DayOfMonth = Day(dtmBirthdate) ' Tag
            .MonthOfYear = Month(dtmBirthdate) ' Monat
            .NoEndDate = True ' Kein Enddatum
        End With
    End With
End For

```

```

        ' Termin/Ereignis speichern
        olAppt.Save
        ' Zähler für gef. Geburtstage erhöhen
        lngHits = lngHits + 1
    End If
Next lngRow
If bolCreated Then
    ' Anwendung verlassen
    objOL.Quit
End If
MsgBox (lngHits) & " Termine/Ereignisse von EXCEL nach OUTLOOK exportiert.", vbInformation,
"Geburtstagstermine"
Exit_Point:
On Error Resume Next
    ' Objektvariablen freigeben
    Set olAppt = Nothing
    Set olPattern = Nothing
    Set olFolder = Nothing
    Set objOL = Nothing
    Set objWS = Nothing
    ' Prozedur sofort verlassen
Exit Sub
Error_Point:
MsgBox "Fehlernummer (" & Err.Number & ") " & Err.Description, vbCritical, "Laufzeitfehler"
Resume Exit_Point
End Sub

```

Listing 2: Geburtstagstermine von EXCEL nach OUTLOOK importieren

10.3 Geburtstagstermine aus OUTLOOK-Kontaktdaten importieren

```

Sub CreateBirthdayEventsFromContacts()
' Aufgabe: Erstellt ein Serienmuster für jeden Kontakt,
' falls ein dazugehöriger Geburtstag hinterlegt ist.
Dim objOL As Object ' Outlook-Anwendung
Dim objContact As Object ' Outlook-Kontakt
Dim olAppt As Outlook.AppointmentItem ' Termin/Ereignis
Dim olCalendar As Outlook.Folder ' Kalender
Dim olContacts As Outlook.Folder ' Kontakte
Dim olPattern As Outlook.RecurrencePattern ' Muster für Serientermine
Dim bolCreated As Boolean ' Statusschalter für Outlook
Dim lngCnt As Long ' Schleifenzähler
Dim lngHits As Long ' Geburtstagszähler
Dim strAge As String ' Lebensalter in Jahren
' Outlook starten
On Error Resume Next
Set objOL = GetObject("Outlook.Application")
If Err.Number <> 0 Then
Err.Clear
Set objOL = CreateObject("Outlook.Application")
bolCreated = True
Else
bolCreated = False
End If
' Fehlerbehandlungsroutine setzen
On Error GoTo Error_Point
' Verzeichnisse setzen
With objOL.GetNamespace("MAPI")
' Geburtstagskalender
Set olCalendar = .GetDefaultFolder(olFolderCalendar).Folders("Geburtstagskalender")
' Kontakte
Set olContacts = .GetDefaultFolder(olFolderContacts)
End With
' Zähler für gültige Geburtstage mit Anfangswert belegen
lngHits = 0
' Outlook-Standardverzeichnis 'Kontakte' mit Zählschleife durchlaufen
For lngCnt = olContacts.Items.Count To 1 Step -1
Set objContact = olContacts.Items(lngCnt)
If objContact.Class = olContact Then
' Prüfen, ob ein gültiges Geburtstagsdatum vorliegt.
If objContact.Birthday <> FormatDateTime("4501-01-01", vbShortDate) Then
' Prüfen, ob der zugehörige Kalendereintrag noch nicht vorhanden ist.
If Not CheckCalEntry(objContact.Birthday, objContact.FirstName, objOL) Then
Set olAppt = olCalendar.Items.Add(olAppointmentItem)
With olAppt
' Ganztätiges Ereignis
.AllDayEvent = True
' Betreff
.Subject = "Geburtstag von " & objContact.FirstName
' Vorlaufzeit in Minuten zwecks Erinnerung an d. Termin/Ereignis
.ReminderMinutesBeforeStart = 1440 ' 24 Stunden
' Schalter zur Erinnerung setzen
.ReminderSet = True
' Kategorie zuordnen
.Categories = "Geburtstag"
' Vertraulichkeit setzen
.Sensitivity = olPrivate
' Lebensalter in Jahren berechnen mit der Funktion 'LengthOfLife'
strAge = LengthOfLife(objContact.Birthday, Now())
.Location = "[Lebensalter: " & strAge & "]"
' Muster für Terminserie festlegen
Set olPattern = .GetRecurrencePattern
End With
With olPattern ' Serienmuster
.RecurrenceType = olRecursYearly ' Häufigkeit: 1-mal jährlich
.PatternStartDate = FormatDateTime(objContact.Birthday, vbShortDate)
.DayOfMonth = Day(objContact.Birthday) ' Tag
.MonthOfYear = Month(objContact.Birthday) ' Monat
.NoEndDate = True
End With
' Termin/Ereignis speichern
olAppt.Save
' Zähler für gef. Geburtstage erhöhen
lngHits = lngHits + 1
End If

```

```

        End If
    End If
Next lngCnt
If bolCreated Then
    ' Anwendung verlassen
    objOL.Quit
End If
MsgBox "Fertig: " & lngHits & " Geburtstage hinzugefügt.", , "Serienereignisse erzeugen"
Exit_Point:
On Error Resume Next
' Objektvariablen freigeben
Set olPattern = Nothing
Set olCalendar = Nothing
Set olContacts = Nothing
Set olAppt = Nothing
Set objContact = Nothing
Set objOL = Nothing
' Procedur sofort verlassen
Exit Sub
Error_Point:
Dim Mldg As String
Mldg = "Fehlernummer (" & Err.Number & "): " & Err.Description
MsgBox Mldg, vbCritical, "Laufzeitfehler"
Err.Clear
Resume Exit_Point
End Sub

```

Listing 3: Serienmuster f. einen Kontakte erstellen, falls dort ein Geburtstagsdatum hinterlegt ist.

```

Function CheckCalEntry(ByVal dtmCheck As Date, strVorname, ByRef objOL As Object) As Boolean
    ' Aufgabe: Prüft, ob ein best. Geburtstagsdatum u. Vorname im Geburtstagskalender vorkommt.
    Dim olItem As Object ' Eintrag im Geburtstagskalender
    Dim olAppt As Outlook.AppointmentItem ' Termin/Ereignis
    Dim olFolder As Outlook.Folder ' Geburtstagskalender
    ' OUTLOOK-Geburtstagskalender Kalender verwenden
    Set olFolder = objOL.Session.GetDefaultFolder(olFolderCalendar).Folders("Geburtstagskalender")
    CheckCalEntry = False
    For Each olItem In olFolder.Items
        If (olItem.Class = olAppointment) Then
            Set olAppt = olItem
            If olAppt.Start = dtmCheck And InStr(olAppt.Subject, strVorname) <> 0 Then
                CheckCalEntry = True
                Exit For
            End If
        End If
    Next olItem
    Set olAppt = Nothing
    Set olFolder = Nothing
    Set olItem = Nothing
End Function

```

Listing 4: Funktion zum Prüfen auf doppelte Geburtstagskalendereinträge

```

Function LengthOfLife(dtmBirthday As Date, Optional dtmToday As Date = 0) As Long
    ' Berechnet das Lebensalter eines Kontakts in Jahren an einem bestimmten Datum, dtmToday.
    ' Wenn dtmToday nicht angegeben ist, wird standardmäßig das heutige Datum verwendet.
    If dtmToday = 0 Then dtmToday = Now()
    LengthOfLife = DateDiff("yyyy", dtmBirthday, dtmToday)
    If DateAdd("yyyy", -LengthOfLife, dtmToday) < dtmBirthday Then
        LengthOfLife = LengthOfLife - 1
    End If
End Function

```

Listing 5: Funktion zur Berechnung des Lebensalters in ganzen Jahren

10.4 Alle Geburtstagstermine aus dem Geburtstagskalender entfernen

```

Sub RemoveBirthdayEvents()
' Aufgabe: Alle Einträge aus dem Geburtstagskalender entfernen
Dim objOL As Object ' Outlook-Anwendung
Dim olAppt As Outlook.AppointmentItem ' Termin/Geburtstag
Dim olItems As Outlook.Items ' Einträge/Geburtstage
Dim olNS As Outlook.Namespace ' Stammobjekt für Verzeichnisse
Dim intLoop As Integer ' Schleifenzähler
Dim bolCreated As Boolean ' Statusschalter für Outlook
Dim intDelete As Integer ' Löschzähler
' Outlook starten
On Error Resume Next
Set objOL = GetObject(, "Outlook.Application")
If Err.Number <> 0 Then
    Err.Clear
    Set objOL = CreateObject("Outlook.Application")
    bolCreated = True
Else
    bolCreated = False
End If
' Fehlerbehandlungsroutine setzen
On Error GoTo Error_Point
' Einträge im Geburtstagskalender bestimmen
Set olNS = objOL.GetNamespace("MAPI")
Set olItems = olNS.GetDefaultFolder(olFolderCalendar).Folders("Geburtstagskalender").Items
olItems.IncludeRecurrences = True
' Schleife über alle Termine im Verzeichnis "Geburtstagskalender"
intDelete = 0
For intLoop = olItems.Count To 1 Step -1
    Set olAppt = olItems.Item(intLoop)
    With olAppt
        If .Class = olAppointment Then
            If .Categories = "Geburtstag" Then
                .Delete
                intDelete = intDelete + 1
            End If
        End If
    End With
Next intLoop
If bolCreated Then
    ' Anwendung verlassen
    objOL.Quit
End If
MsgBox "Fertig!" & Str(intDelete) & " Einträge entfernt!", vbInformation, "Geburtstagsdaten entfernen"
Exit_Point:
On Error Resume Next
' Objektvariablen freigeben
Set olItems = Nothing
Set olNS = Nothing
Set objOL = Nothing
' Prozedur sofort verlassen
Exit Sub
Error_Point:
MsgBox "Laufzeitfehler (" & Err.Number & ") " & Err.Description, vbCritical, "Geburtstagsdaten entfernen"
Resume Exit_Point
End Sub

```

Listing 6: Alle Einträge aus dem OUTLOOK-Geburtstagskalender entfernen

10.5 Geburtstagsgrüße erzeugen

```

Sub BirthdayGreetings ()
' Aufgabe: Geburtstagsgrüße erzeugen
Dim objOL As Outlook.Application ' Anwendung
Dim objContact As Outlook.ContactItem ' Kontakte
Dim objMail As Outlook.MailItem ' E-Mail-Nachricht
Dim objTemplate As Outlook.MailItem ' Entwurf einer E-Mail-Nachricht
Dim olDrafts As Outlook.MAPIFolder ' Verzeichnis f. Entwürfe
Dim olItems As Outlook.MAPIFolder ' Verzeichnis f. Kontakte
Dim objNS As Outlook.Namespace ' Stammobjekt für Datenquellen
Dim objItem As Object ' Kontakt
Dim bolCreated As Boolean ' Statusanzeiger f. Outlook
Dim intCnt As Integer ' Zahl d. erzeugten E-Mail-Entwürfe
Dim dtmDate As Date ' Datum als Date-Datentyp
Dim strBirthday As String ' Monat-Tag des Geburtstagsdatums
Dim strDate As String ' Monat-Tag des Eingabedatums
Dim varDate As Variant ' Eingabedatum als Variant-Datentyp

' Fehlerbehandlungsroutine
On Error GoTo Error_Point

' Geburtstagsdatum eingeben
varDate = InputBox("Bitte das Geburtstagsdatum eines Kontakts eingeben!", "Format: tt.mm.jjjj")
If Not IsDate(varDate) Then
    MsgBox "Abbruch: Kein gültiges Datum!"
    GoTo Exit_Point
End If
' varDate in den Datentyp 'Datum' konvertieren
dtmDate = CDate(varDate)
' Datum zusammensetzen aus Monat und Tag
strDate = Month(dtmDate) & "-" & Day(dtmDate)

' Outlook starten
On Error Resume Next
Set objOL = GetObject(, "Outlook.Application")
If Err.Number <> 0 Then
    Err.Clear
    Set objOL = CreateObject("Outlook.Application")
    bolCreated = True
Else
    bolCreated = False
End If

Set objNS = objOL.GetNamespace("MAPI")
With objNS
    Set olDrafts = .GetDefaultFolder(olFolderDrafts) ' Entwürfe
    Set olItems = .GetDefaultFolder(olFolderContacts) ' Kontakte
End With
For Each objItem In olItems.Items
    If TypeOf objItem Is Outlook.ContactItem Then
        Set objContact = objItem
        With objContact
            strBirthday = Month(.Birthday) & "-" & Day(.Birthday)
            If strBirthday = strDate Then
                If .FullName <> "Volker Thormählen" Then
                    ' Neue E-Mail erzeugen
                    Set objMail = objOL.CreateItem(olMailItem)
                    With objMail
                        ' Betreff
                        .Subject = "Geburtstagsgrüße"
                        ' Empfänger (... mit der ersten E-Mail-Adresse des Kontakts)
                        .Recipients.Add objContact.Email1DisplayName
                        ' Textkörper
                        .Body = "Hallo " & objContact.FirstName & ","
                        ' Entwurf suchen
                        Set objTemplate = olDrafts.Items.Find("[Subject] = ""Geburtstagsgrüße""")
                        If Not objTemplate Is Nothing Then
                            ' Entwurf gefunden
                            .Body = .Body & objTemplate.Body
                            ' .Body = objTemplate.Body
                        Else
                            ' Entwurf nicht gefunden
                            .Body = "Herzlichen Glückwunsch zum Geburtstag!"
                        End If
                    End With
                    intCnt = intCnt + 1
                End If
            End If
        End With
    End If
End For
Exit_Point:
End Sub

```

```

        ' Speichert die Nachricht im aktuellen Ordner
        .Save ' .Close(olSave) sendet die Nachricht
    End With
End If
End If
End With
End If
Next objItem
If intCnt > 0 Then
    MsgBox "Fertig!" & Str(intCnt) & " E-Mail-Entwurf erzeugt.", _
        vbInformation, "Geburtstagsgrüße ausgeben"
Else
    MsgBox "Fertig! Keinen E-Mail-Entwurf erzeugt.", _
        vbInformation, "Geburtstagsgrüße ausgeben"
End If
If intCnt > 0 Then
    MsgBox "Fertig!" & Str(intCnt) & " E-Mail-Entwurf erzeugt.", vbInformation, "Geburtstagsgrüße ausgeben"
Else
    MsgBox "Fertig! Keinen E-Mail-Entwurf erzeugt.", vbInformation, "Geburtstagsgrüße ausgeben"
End If
Exit_Point:
On Error Resume Next
If bolCreated Then
    objOL.Quit
End If
' Objektvariablen freigeben
Set objOL = Nothing
Set objContact = Nothing
Set objMail = Nothing
Set objTemplate = Nothing
Set olDrafts = Nothing
' Prozedur sofort verlassen
Exit Sub
Error_Point:
MsgBox "Laufzeitfehler (" & Err.Number & ") " & Err.Description, vbCritical, "Geburtstagsgrüße ausgeben"
Resume Exit_Point
End Sub

```

Listing 7: Geburtstagsgrüße erzeugen

11 Verzeichnisse

11.1 Abbildungen

Abbildung 1: Verweise setzen zur Office Automation	1
Abbildung 2: Schaltfläche zum Verschieben von Kalenderelementen	2
Abbildung 3: Erfolgsmeldung nach dem Verschieben von Serienterminen.....	2
Abbildung 4: Geburtstagskalender in OUTLOOK neu erstellen	3
Abbildung 5: Schaltfläche Kategorisieren im OUTLOOK-Menüband	3
Abbildung 6: Neue Kategorie 'Geburtstag' hinzufügen	3
Abbildung 7: Geburtstagstermine in EXCEL-Tabellenblatt	4
Abbildung 8: Geburtstermine in OUTLOK-Geburtstagskalender eintragen	4
Abbildung 9: OUTLOOK-Kontaktdaten von Volker Thormählen	5
Abbildung 10: Details eines OUTLOOK-Kontakts erfassen.....	5
Abbildung 11: Feld zur Aufnahme des Geburtstags des zuvor gezeigten OUTLOOK-Kontakts	6
Abbildung 12: Geburtstage in OUTLOOK-Kontakte finden und in den Geburtstagskalender einstellen	6
Abbildung 13: Geburtstagskalender öffnen	6
Abbildung 14: Liste der Einträge im OUTLOOK-Geburtstagskalender	7
Abbildung 15: Terminserie für den OUTLOOK-Kontakt 'Volker Thormählen'	7
Abbildung 16: Erfolgsmeldung bei Entfernung aller Termine aus dem Geburtstagskalender	8

11.2 Listings

Listing 1: Serientermine in den Geburtstagskalender verschieben	II
Listing 2: Geburtstagstermine von EXCEL nach OUTLOOK importieren	IV
Listing 3: Serienmuster f. einen Kontakte erstellen, falls dort ein Geburtstagsdatum hinterlegt ist. ...	VI
Listing 4: Funktion zum Prüfen auf doppelte Geburtstagskalendereinträge	VI
Listing 5: Funktion zur Berechnung des Lebensalters in ganzen Jahren	VI
Listing 6: Alle Einträge aus dem OUTLOOK-Geburtstagskalender entfernen	VII

11.3 Literatur

- [1] S. Zhang, „Auto Move Birthday Events to a Specific Calendar,“ 31 08 2017. [Online]. Available: <https://www.datanumen.com/blogs/auto-move-birthday-events-specific-calendar-outlook/>. [Zugriff am 09 08 2019].